United States
Department of
Agriculture

Forest Service

**Pacific Southwest
Forest and Range
Experiment Station**

General Technical
Report PSW-112

MIX: A Computer Program to Evaluate Interaction Between Chemicals

Jacqueline L. Robertson **Kimberly C. Smith**

Robertson, Jacqueline L.; Smith, Kimberly C. 1989. **MIX: a computer program to evaluate interaction between chemicals.** Gen. Tech. Rep. PSW-112. Berkeley, CA: Pacific Southwest Forest and Range Experiment Station, Forest Service, U.S. Department of Agriculture; 9 p.

A computer program, MIX, was designed to identify pairs of chemicals whose interaction results in a response that departs significantly from the model predicated on the assumption of independent, uncorrelated joint action. This report describes the MIX program, its statistical basis, and instructions for its use.

Retrieval Terms: mixtures, synergism, antagonism

The Authors:

JACQUELINE L. ROBERTSON, a research entomologist, is in charge of the unit developing improved technology for integrated management of western forest insects, with headquarters in Berkeley, Calif. **KIMBERLY C. SMITH** was formerly a mathematical statistician with the Station.

Publisher:

Pacific Southwest Forest and Range Experiment Station
P.O. Box 245, Berkeley, California 94701

June 1989

MIX: A Computer Program to Evaluate Interaction Between Chemicals

Jacqueline L. Robertson **Kimberly C. Smith**

CONTENTS

INTRODUCTION

In the absence of biological data, elaborate theories and statistical procedures have been developed to explain the interaction of chemicals applied in a mixture to a target organism. Since Bliss (1939) first described statistical tests of joint action, complex methods have been devised to describe various hypothetical types of chemical interaction (e.g., Plackett and Hewlett 1948, Hewlett and Plackett 1950). These methods have not been widely used in entomological investigations because they are mathematically complex, difficult to interpret, and generally irrelevant to the practical consideration of identifying mixtures that are significantly more effective together than one would expect from their individual toxicities.

For general purposes, only three types of joint action between chemicals need be defined: synergism, antagonism and independence. We define synergism as activity significantly greater than that expected from the toxicities of the separate mixture components. Antagonism is activity that is significantly less than expected. Independence occurs when activity of the mixture does not differ significantly from that predicted on the basis of effectiveness of the separate components.

We designed a computer program called MIX to identify pairs of chemicals whose interaction results in a response that departs significantly from those predicted by the model of independent, uncorrelated joint action (Bliss 1939). In this model, toxicity of one component is assumed to be unaffected by the presence of the other chemical, and susceptibility to one chemical is not correlated with susceptibility to the other. Rejection of the model may indicate some correlation of susceptibilities, non-independent interaction, or both. The statistical procedure used in MIX is flexible: either a response range or a single response level may be evaluated.

This report describes the MIX computer program, its statistical basis, and instructions for its use.

STATISTICAL MODEL

The statistical model used in MIX was described by Robertson and Smith (1984) as follows:

Under the null hypothesis of independent, uncorrelated action, the probability $(1 - P_0)$ of surviving an exposure to a chemical mixture comprised of concentration X_1 of chemical 1 and X_2 of chemical 2 is the product of the probability of survival from natural causes (1- m), the probability of surviving X_1(1-P_1) and the probability of surviving $X_2(1 - P_2)$. Hence,

$$1\text{-}P_0 = (1\text{-}m)(1\text{-}P_1)(1\text{-}P_2)$$

or

$$P_0 = m + \{(1\text{-}m)[P_1 + (1\text{-}P_1)P_2]\}$$

If the true probability of response to the mixture is denoted by P, then the null hypothesis can be expressed as H_0: $P = P_0$.

For an actual experiment with a mixture of two chemicals, a dose X of the mixture is made up of X_1 of chemicals 1 and X_2 of chemical 2 in accordance with the mixing proportions used. Of n insects exposed to the mixture at concentration X, an observed proportion, $\hat{P}$, is killed. The expected response rate under the null hypothesis, P_0, is estimated as

$$\hat{P}_0 = \hat{m} + (1 - \hat{m})[\hat{P}_1 + (1 - \hat{P}_1)\hat{P}_2]$$

where m is the proportional mortality of the n_0 subjects in the control group, and $\hat{P}_1$ and $\hat{P}_2$ are the predicted response probabilities to X_1 and X_2, adjusted for natural mortality and estimated from the probit regression lines for chemicals 1 and 2 separately.

The difference $(\hat{P} - \hat{P}_0)$ is standardized by an estimate of the inherent variability associated with P and P_0 to assure that the magnitude of the difference is not the result of chance alone. Because $\hat{P}$ and $\hat{P}_0$ are statistically independent,

$$\begin{aligned}\mathrm{Var}(\hat{P} - \hat{P}_0) &= \mathrm{Var}(\hat{P}) + \mathrm{Var}(\hat{P}_0) \\ &= P(1\text{-}P)/n \\ &\quad + (1 - P_0)^2[1 + C_m)(1 + C_1)(1 + C_2) - 1]\end{aligned}$$

where $C_m = \mathrm{Var}(\hat{m})/(1 - m)^2$, $C_1 = \mathrm{Var}(\hat{P}_1)/(1 - \hat{P}_1)^2$ and $C_2 = \mathrm{Var}(\hat{P}_2)/(1 - P_2)^2$. Under the null hypothesis, this variance can be estimated by

$$\begin{aligned}\text{estimated } \mathrm{Var}(\hat{P} - \hat{P}_0) &= \hat{P}_0(1 - \hat{P}_0)/n \\ &\quad + (1 - \hat{P}_0)^2[(1 + \hat{C}_m)(1 + \hat{C}_1)(1 + \hat{C}_2) - 1]\end{aligned}$$

Calculations of $\hat{C}_i$ require the estimated intercept $\hat{\beta}_{1i}$ slope $\hat{\beta}_{2i}$, and their estimated variances and covariance, $\hat{\sigma}_{1i}, \hat{\sigma}_{2i}, \hat{\sigma}_{12i}$ from the prohibit analysis of chemical i alone, i = 1,2. Then $Y_i = \beta_{1i} + \beta_{2i} \log_{10}(X_i)$, $V_i = \hat{\sigma}_{1i} + 2\hat{\sigma}_{12i} \log_{10} X_i + \hat{\sigma}_{2i} \log^2_{10}(X_i)$ and $C_i = \exp(-y^2_i)V_i/[2\pi(1 - \hat{P}_i)^2]$; $\hat{C}_m = \hat{m}/[n_0(1 - \hat{m})]$.

The test statistic at dose X is $Z^2(X) \dfrac{(P - P_0)^2}{\text{estimated } \mathrm{Var}(\hat{P} - \hat{P}_0)}$

Under H_0, if the number of observations in the control group at dose X of the mixture and at all doses of the individual components are all large, the central limit theorem assures that $Z^2(X)$ will be well approximated by a χ^2 distribution with one degree of freedom [$\chi^2_{(1)}$]. As long as all doses are located to avoid very high (>90 pct) and very low (<10 pct) response probabilities, sample sizes of 30 per dose should suffice for purposes of the approximation. The null hypothesis is rejected by an α-level test if $Z^2(X)$ exceeds the (1 -α) 100 percent quantile of the $\chi^2_{(1)}$ distribution.

As an approximate procedure for evaluating the toxicity of a mixture over a range of doses, the $Z^2(X)$ are summed over the k concentrations lying in the range of interest,

$$Z^2 = \sum_{j=1}^{k} Z^2(X_j)$$

The hypothesis of independent, uncorrelated action by the mixture over the entire range is rejected if Z^2 exceeds the (1-α) 100 percent quantile of the χ^2 distribution with k degrees of freedom.

DATA INPUT FILE

Parameter estimates from statistical analyses with each component and the mixture are placed in a data file for use by MIX. The file name is arbitrary, but must include 10 or fewer characters and no blanks. The name of the file is MIXIN (*fig. 1*).

Line 1 of the data file identifies the first chemical in the mixture (the choice is arbitrary, since either chemical may be considered number 1). Line 17 (*fig. 1*) is another example for a second data set. This identifier, a maximum of 10 characters, must be typed within single quotation marks (e.g., 'permethrin').

Line 2 (*fig. 1*) lists seven parameter estimates or data related to the insects' response to chemical 1. Line 18 is another example for a second data set. The values must be separated by one or more blank spaces and must be entered in specific order. Parameter estimates are available from the printout of a probit or logit analysis program such as POLO (Russell and others 1977). In *figure 2*, the parameters are labelled as:

BI: estimated intercept
B2: estimated slope
S1: estimated variance of B 1
S2: estimated variance of B2
S12: estimated covariance of BI and B2
ADJ: heterogeneity adjustment factor
PROP: Proportion (as a decimal) of chemical 1

The next line (lines 3 and 19) identifies the second chemical in the mixture, entered as described for the first chemical. In the next line (lines 4, 20), the seven parameter estimates or data for the second chemical in the mixture are entered in the same format as values for the first chemical.

Data for the actual bioassay done with the mixture of chemical 1 and chemical 2 follow. The three entries (K, EXP, DEAD) in the first data line (lines 5, 21) show the number of doses or concentration levels tested (K) not counting the control group(s), number tested in the control group (EXP), and the number dead

```
1.   'permethrin'
2.   2.53 5.09 .1097 .6959 .2483 1.1825 .1
3.   'dimilin'
4.   1.43 1.47 .015 .0392 .0156 2.2685 .9
5.   11 69 13
6.   .001 10 6
7.   .01 10 5
8.   .1 68 29
9.   .2 60 38
10.  .3 60 48
11.  .5 59 48
12.  .7 60 53
13.  1. 70 69
14.  2. 60 60
15.  3. 60 60
16.  10. 10 10
17.  'permethrin'
18.  2.66 4.77 .0695 .2580 .1219 1.4107 .1
19.  'methoprene'
20.  1.4 1.64 .0134 .0367 .0151 1.6960 .9
21.  10 68 10
22.  .001 10 0
23.  .01 68 17
24.  .02 60 19
25.  .03 60 22
26.  .05 60 26
27.  .07 57 38
28.  .1 70 51
29.  .5 60 56
30.  1. 70 64
31.  10. 10 10
```

Figure 1--Contents of a sample input file named 'MIXIN.'

in the control group (DEAD). Following this line, data for each dose or concentration of the mixture are entered. Values listed in each line (6-10 and 21-31, *fig.1*) are the dose (concentration) in actual units tested (DOSE), number of subjects exposed to the dose (EXP), and number of deaths observed among the subjects tested (DEAD). The format is the same for all of these entries; the three entries on each line must be in the order listed above and values must be separated by one or more blank spaces. The data file needs no special character to identify the end of the file.

RUNNING THE MIX PROGRAM

The MIX program (see *Appendix*) is written in Fortran 77. It consists of a main routine (MIXTURES. F77) and a subroutine (PINORM. 77). PINORM. 77 computes (X) for any X. All computations in both routines are done in double precision. The executable routine is MIXTURES.PR this program should be copied or moved into the directory that contains your input file.

After entering the directory where MIXTURES.PR and the input file are located, type X MIXTURES and hit the return key. This message will appear:

ENTER FILE NAME WHERE DATA IS STORED.

The name of the input file should be typed at the end of this line; the name must be placed within single quotes. Hit the return key. For example, the completed line for the sample data file MIXIN would appear as

ENTER FILE NAME WHERE DATA IS STORED. `MIXIN'

Output from MIX will be stored in the working directory. This file is created by the program if it does not already exist. On the next line, you will be asked to supply the name of the output file. The prompt line that appears is

ENTER FILE NAME TO STORE OUTPUT.

The name chosen should contain no more than 10 characters and no blanks (e.g., 'MIXOUT'). The name should be typed and

```
DIET TEST. PERMETHRIN + METHOPRENE


INTERCEPTS AND SLOPES UNCONSTRAINED.  PREPARATION IS (1) PERMETHRIN
ESTIMATING NATURAL RESPONSE

MAXIMUM LOG-LIKELIHOOD   -143.30810

          B1
                PARAMETER        STANDARD ERROR      T RATIO
PERMETHR      2.6591623          .26359936         10.087894
NATURAL        .58769126-01      .20471423-01       2.8707878
SLOPE         4.7721107          .50788916          9.3959688
          B2
VARIANCE-COVARIANCE MATRIX
              S1   PERMETHR       NATURAL         SLOPE
  PERMETHR       .6948462-01     .6208465-03    .1219397<------- S12
  NATURAL        .6208465-03     .4190794-03    .2681468-02
  SLOPE          .1219397        .2681468-02    .257914<-------- S2

CHI-SQUARED GOODNESS OF FIT TEST

PREPARATION   SUBJECTS     RESPONSES   EXPECTED     DEVIATION     PROBABILITY
PERMETHR        10            1.          .588         .412          .058769

CHI-SQUARE   38.089        DEGREES OF FREEDOM  27      HETEROGENEITY   1.4107<--ADJ
```

Figure 2--POLO output from analysis of bioassay data with individual chemicals.



Figure 3--Copy of output stored in the file named 'MIXOUT.'

entered, then hit the return key. For this example, the complete line would be

ENTER FILE NAME TO STORE OUTPUT. 'MIXOUT'

Upon receipt of these file names, MIX will process the data in the input files, then place the results into the output file. When the process is complete, STOP will appear on the screen.

To view the contents of the output file, enter TYPE followed by at least one space and the name of the output file. For the example shown in this manual, the complete line would be

TYPE MIXOUT

Hard copy of this output is obtained by typing QPRINT followed by at least one space and the name of the output file.

SAMPLE OUTPUT

Sample input data listed in *figure 1*, were subjected to probit analysis (*fig. 2*). Relevant output was stored in a file called MIXOUT (*fig. 3*). Data and statistical parameters entered for the individual chemicals are listed in lines 1-8 for permethrin and dimilin and lines 24-31 for permethrin and methoprene.

Next, MIX calculations of expected mortality and χ^2 values for each dose calculated as described in section 1 (Statistical Model) for the null hypothesis of independent joint action are listed. In this sample output, these calculations appear in the last two columns of lines 10-22 and 34-44. The total χ^2 with the appropriate degrees of freedom (number of doses [including the control]-1) are listed next (line 23, line 45). To determine whether significant departures from the null hypothesis level of choice (e.g., P=0.05) can be located in a χ^2 table available in any statistical text. If the value calculated by MIX is less than the tabular value, the null hypothesis cannot be rejected. If the χ^2 value is greater than the tabular value, than the hypothesis of independent joint action is rejected.

A plot of the data should be prepared to determine the direction of significant departures of the results from values predicted by the hypothesis. Ideally, all the data points will lie to the left (significant antagonism) or the right (significant synergism) of expected values. However, points on either side of the expected line occur (Robertson and Smith 1984): in these instances, χ^2 values above and below 50 percent may be examined to assist in interpretation of the results. Degrees of freedom (lines 23, 45) are equal to the number of doses (including controls) minus 1; if a subset of data is considered, degrees of freedom are calculated in the same way. The calculated χ^2 value (lines 23, 45) should be compared with tabular value at the appropriate degrees of freedom and probability value of choice to determine significance of the results.

APPENDIX: PROGRAM LISTING

```
c MIXTURES.F77 CONDUCTS A CHISQUARE GOODNESS OF FIT TEST OF THE
c HYPOTHESIS OF INDEPENDENT, UNCORRELATED JOINT ACTION BY COMPONENTS
c IN A CHEMICAL MIXTURE. IT IS ASSUMED THAT THE MIXTURE IS COMPRISED OF
c TWO COMPONENTS IN A FIXED MIXING RATIO. IT IS ALSO ASSUMED THAT A
c PROBIT MODEL ADEQUATELY DESCRIBES THE DOSE-RESPONSE RELATIONSHIP FOR
c EACH OF THE TWO CHEMICALS IF APPLIED ALONE. FOR A DESCRIPTION OF THE
c STATISTICAL ASPECTS OF THE PROCEDURE, SEE ROBERTSON AND SMITH,
c "JOINT ACTION OF PYRETHROIDS WITH ORGANOPHOSPHORUS AND CARBAMATE
c INSECTICIDES,...", J.ECON.ENT.,FEBRUARY,1984. THE NOTATION USED IN
c THE PROGRAM FOLLOWS THAT USED IN THE PAPER, WITH A FEW NECESSARY
c EXCEPTIONS.
c
c
c COMPANION PROGRAM IS PINORM.F77, A PROGRAM WHICH COMPUTES THE CDF OF
c A STANDARD NORMAL VARIABLE.
c
c
        implicit double precision(a-h,o-z)
        parameter(pi=3.141592654)
        character*10 file_in,file_out,chemical(2)
        dimension b1(2),b2(2),s1(2),s2(2),s12(2),adj(2),prop(2),
     1  x(2),y(2),p(2),v(2),c(2)
c
        print *, 'enter file name where data is stored. '
        read *, file_in
        open(3,file=file_in)
        print *, 'enter file name to store output. '
        read *, file_out
        open(4,file=file_out)
c
 1000   do 1 i=1,2
          read(3,*,end=9999) chemical(i)
          read(3,*) b1(i),b2(i),s1(i),s2(i),s12(i),adj(i),prop(i)
   1    continue
c
        write(4,10) chemical,prop,b1,b2,s1,s2,s12,adj
   10   format (1x,'chemical       ',a10,3x,a10/,
     1           ' proportion     ',2f10.4,/,
     2           ' intercept b1   ',2f10.4,/,
     3           ' slope b2       ',2f10.4,/,
     4           ' var(b1)        ',2f10.4,/,
     5           ' var(b2)        ',2f10.4,/,
     6           ' covar(b1,b2)   ',2f10.4,/,
     7           ' het adjust     ',2f10.4)
c
        read(3,*) k, exp, dead
        control=dead/exp
        cc=control/((1.-control)*exp)
c
        write(4,20) chemical,exp,control
   20   format(1x,'mixture of ',a10,' and 'a10,/,
     1           '    dose   no_exp obs.mort  exp.mort   chisq   ',/,
     2          ' controls  ',f6.0,f10.4)
        bigzsq=0.
        do 3 j=1,k
```

```
            read(3,*) dose,exp,dead
            phat=dead/exp
            do 2 i=1,2
                  x(i)= prop(i)*dose
                  y(i)=b1(i)+b2(i)*dlog10(x(i))
                  p(i)=pinorm(y(i))
                  v(i)=s1(i)+2.*s12(i)*dlog10(x(i))+s2(i)*dlog10(x(i))**2
                  c(i)=dexp(-y(i)**2)*v(i)*adj(i)/(2.*pi*(1.-p(i))**2)
   2        continue
            p0=control + (1.-control)*(p(1) + (1.-p(1))*p(2))
            var = phat*(1.-phat)/exp +
     1            (1.-p0)**2*((1.+cc)*(1.+c(1))*(1.+c(2)) - 1.)
            zsq=(phat-p0)**2/var
            write(4,30) dose,exp,phat,p0,zsq
   30       format(1x,f10.4,f6.0,3f10.4)
            bigzsq=bigzsq+zsq
    3     continue
c
          write(4,40) bigzsq,k
   40     format(1x,'chisq= ',f10.4,' with ',i4,' degrees of freedom'///)
c
          go to 1000
c
  9999    stop
          end



          DOUBLE PRECISION FUNCTION PINORM(Z)
C
C         THIS IS AN ADAPTATION ON THE SUBROUTINE PINORM PROGRAMMED
C         BY ROBERT RUSSELL AT FORT COLLINS.
C         THIS ROUTINE DOES NOT HAVE ENTRY POINTS FOR ERF AND ERFC
C
C         THE PROGRAM COMPUTES THE CUMULATIVE PROBABIILITY FUNCTION
C         OF THE UNIT NORMAL DISTRIBUTION AT Z.
C
C         THE METHOD IS TO COMPUTE THE COMPLEMENTARY ERROR FUNCTION,
C         ERFC AND USE THE IDENTITY:  P(Z)=1 - .5*ERFC(Z/SQRT(2)).
C
C         FOR ABS(Z)/SQRT(2) GREATER THAN .47, USE HART'S RATIONAL
C         APPROXIMATION TO ERFC, THEN IDENTITY ABOVE.
C         FOR ABS(Z)/SQRT(2) LESS THAN .47, USE TAYLOR SERIES APPROX.
C         TO ERF, THEN ERFC=1-ERF, THEN IDENTITY ABOVE.
C
C         FINALLY, IF Z<0, P(Z)=1-P(-Z).
C
C         KS, DEC. 80
C
          IMPLICIT DOUBLE PRECISION (A-Z)
          DIMENSION A(4),B(5)
          DATA A/10.0046412, 8.42655286, 3.46025933, .562353612/
          DATA B/10.0046412, 19.7155807, 15.7022881, 6.09074879, 1./
          X=DABS(Z)/DSQRT(2.D0)
C
          IF (X.GT..47) GO TO 20
C
C                 IF X<.47, USE TAYLOR SERIES APPROXIMATION.
C
```

```
      W=X**2
      E=X*(1.-W*((1./3.)-W*.5*((1./5.)-W*(1./3.)*((1./7.)-W*.25*
     $ ((1./9.)-W*(1./55.) )))))
      E=1.-1.12837917*E
      GO TO 40
C
C             FOR X>.47 BUT X<5. USE HART.
C
   20 IF (X.GT.5.) GO TO 30
      AA=A(1)+(A(2)+(A(3)+A(4)*X)*X)*X
      BB=B(1)+(B(2)+(B(3)+(B(4)+B(5)*X)*X)*X)*X
      E=EXP(-X**2)*AA/BB
      GO TO 40
C
C             FOR X>5. RETURN ERFC=0.
C
   30 E=0.
C
C             ADJUST FOR NEGATIVE ARGUMENTS.
C
   40 IF (Z.LT.0.) E=2.-E
      PINORM=1.-(.5*E)
      RETURN
      END
```

REFERENCES

Bliss, Chester I. 1939. **The toxicity of poisons applied jointly.** Annals of Applied Biology 26: 585-615.

Hewlett, P. S.; Plackett, R. L. 1950. **Statistical aspects of the independent joint action of poisons, particularly insecticides. II. Examination of data for agreement with the hypothesis.** Annals of Applied Biology 37: 527-552.

Plackett, R. L.; Hewlett, P. S. 1948. **Statistical aspects of the independent joint action of poisons, particularly insecticides. I. The toxicity of a mixture of poisons.** Annals of Applied Biology 35: 347-358.

Robertson, Jacqueline L.; Smith, Kimberly C. 1984. **Joint action of pyrethroids with organophosphorus and carbamate insecticides applied to western spruce budworm (Lepidoptera: Tortricidae).** Journal of Economic Entomology 77: 16-22.

Russell, Robert M.; Robertson, Jacqueline L.; Savin, N. E. 1977. **POLO: a new computer program for probit analysis.** Bulletin of the Entomological Society of America 23: 209-213.